
modélisation de haut 
niveau

implantation

? 

ProblémaAque 

•  Approches formelles de descripAon 
  SCADE, SDL, réseaux de Petri, etc. 

  couplés à des générateurs de code 

  mise en place de la réparAAon 

•  Eléments concrets de mise en œuvre 

  réseaux, intergiciels, etc. 

  contraintes sur l’applicaAon réparAe 

  adaptaAon aux besoins de l’applicaAon 

•  Comment rassembler ces deux aspects? 

  passer de la descripAon formelle à 
l’implantaAon 

  évaluer les performances de l’applicaAon 
o  temps d’exécuAon 
o  place en mémoire 
o  fiabilité 
récepteur

émetteur
Transmet une donnée
de type A à

Les langages de descripAon d’architecture (ADL) ‐ 1 

•  La concepAon des systèmes devient complexe 
  difficulté de compréhension du système 

  preuve de cohérence de l’architecture du système 

•  Un formalisme de descripAon est nécessaire 
  pour décrire l’architecture 
  pour analyser et vérifier ses propriétés 

•  3 éléments principaux : 
  les composants 

  les liens entre les composants 

  une sémanAque associée à ces composants 
Les langages de descripAon d’architecture (ADL) ‐ 2 

•  ADL formels 
 Formalisent la descripAon du foncAonnement d’un système 
 S’intègrent mal dans une démarche de généraAon automaAque 

 Wright, Rapide 

•  ADL concrets 
 Décrivent l’architecture afin de la générer automaAquement 
 UML 2, AADL 

•  ADL restreints 
 Décrivent l’assemblage de composants logiciels ou la cohérence 
de l’applicaAon 

 ArchJava, Fractal 
Architecture Analysis & Design Language 

•  Anciennement « Avionics Architecture DescripAon Language » 
•  EvoluAon de MetaH, qui était développé par Honeywell 

  Dédié aux systèmes réparAs embarqués temps‐réel 
  Décrit des éléments matériels et logiciels 
  Conçu pour permecre la généraAon de systèmes exécutables 

o  expression des caractérisAques des éléments du système 
o  traducAon en langage de programmaAon 

•  Plusieurs représentaAons 
  representaAon textuelle 

o  pour contrôler tous les détails du système 
  représentaAon en XML 

o  pour l’interopérabilité entre ouAls 
  représentaAon graphique 

o  convenable pour avoir une vision globale du système 
  profil UML 2, et représentaAon en UML 1.4 

•  Version 1.0 publiée en 2004 
•  Version 2 sera publiée en 2009 
AADL 1.0 (hcp://www.aadl.info) 

•  Publié en 2004 
•  Déjà uAlisé par de grands projets 

  COTRE (Airbus) 
  ASSERT (ESA, EADS, ENST, INRIA, etc.) 
  Topcased (Airbus, CNES, ENST, etc…) 
  … 

•  Quelques ouAls disponibles 
  OSATE : ouAl de référence pour Eclipse (SEI/CMU) 

o  syntaxe textuelle et graphique 
o  vérificaAons syntaxiques et sémanAques générales 
o  plusieurs extensions pour la vérificaAon d’architectures 

  STOOD : ouAl de modélisaAon basé sur la méthode HOOD (Ellidiss) 
o  ouAl graphique (syntaxe UML & HOOD) 
o  générateur de code pour des applicaAons monolithiques 

  Ocarina : suite d’ouAls pour générer des applicaAons (ENST) 
o  noyau central pouvant être intégré dans différentes applicaAons 
o  plusieurs générateurs d’applicaAons (Ada et C) 
Principes généraux 

•  Composants reliés entre eux 
  Connexions reliant les interfaces (ports, paramètres…).  
  Chaque composant, défini par un type + une ou plusieurs implémentaAons 

•  Les composants pouvant contenir des sous‐composants 
•  Possibilité d’acribuer des propriétés à chaque élément 

  composant, sous‐composant, connexion, port… 

•  CaractérisAques de l’implémentaAon pouvant varier 
  selon des modes de configuraAon 
  selon les propriétés 

•  But: modéliser la réalité 
  générer facilement un système foncAonnel à parAr de sa descripAon.  
  Tous les éléments correspondent à quelque chose de concret 

•  AADL: un langage descripAf 
   les éléments peuvent être donnés dans n’importe quel ordre 
Catégories de composants 

•  Éléments de base d’une descripAon architecturale 
• DéclaraAon en plusieurs parAes 

 component type : interface 
 component implementaAon : structure interne 
 un type peut avoir plusieurs implémentaAons 

•  Plusieurs catégories 
 matériels (execu%on pla,orm components) 

o processor, memory, bus, device 

 logiciels (so1ware components) 
o thread, data, process, thread group, subprogram 

 systèmes (system composi%on) 
o system 
Les composants matériels 

•  Plusieurs catégories 
 processor : micro‐processeur + ordonnanceur 
 memory : disque dur, mémoire vive, etc. 

 bus : réseau, etc. 
 device : composant dont on ignore la structure interne 

• Un processor modélise processeur + noyau contenant 
entre‐autres un ordonnanceur. 

• Un device sert typiquement à modéliser un capteur 

Device Memory bus Processor 

Les composants logiciels 

•  Plusieurs catégories 
 thread : fil d’exécuAon (ou thread dans les noyaux) 
 data : structure de données 

 process : processus, un espace mémoire pour l’exécuAon des 
threads qu’il conAent 

 thread group : crée une hiérarchie dans les threads 

 subprogram : procédure, comme pour les langages de 
programmaAon. N’as pas de valeur de retour 

• Un process doit contenir au moins un thread. 
Thread data Threadgroup process subprogram 

Les systèmes 

•  Permecent de structurer la descripAon. 

•  ConAennent les composants qui peuvent être manipulés de 
façon indépendante : 
 system 
 processor, memory, device, bus 

 process, data 

• MAIS PAS : 
 thread 
 thread group 

 subprogram 

System 

data data

thread data

thread group data, thread, thread group

process thread, thread group

processor memory

memory memory

system tous sauf subprogram, thread et thread group

Les sous‐composants 

• Un composant peut avoir des sous‐composants 
 décrits dans les implémentaAons des composants 

• Une modélisaAon AADL est une arborescence de 
composants 
Exemple de composants 

thread execution_thread
end execution_thread;

process a_process
end a_process;

process implementation a_process.one_thread
subcomponents
  thread1 : thread execution_thread;
end a_process.one_thread;

process implementation a_process.two_threads
subcomponents
  thread1 : thread execution_thread;
  thread2 : thread execution_thread;
end a_process.two_threads;

subprogram spg1 end spg1;
subprogram spg2 end spg2;
thread a_thread end a_thread;

thread implementation a_thread.example
calls {
  call1 : subprogram spg1;
  call2 : subprogram spg2;
  call3 : subprogram spg1;
};
end a_thread.example;

Appels de sous‐programmes 

•  Une implantaAon de sous‐programme ou de thread peut contenir des séquences 
d’appels à des sous‐programmes 

•  L’ordre des appels est important 

•  DescripAon du flux d’exécuAon dans les composants 
Type 1

Type 2

Impl 1A

Impl 1B

Type 3

Impl 3A

Impl 3B

Impl 3C

Extension des composants 

•  un composant AADL peut étendre un autre 
(component extension) 

Extension

Implémentation

thread execution_thread
end execution_thread;

process a_process
end a_process;

process implementation a_process.one_thread
subcomponents
  thread1 : thread execution_thread;
end a_processus.one_thread;

process implementation a_process.two_threads
extends a_process.one_thread
subcomponents
  thread2 : thread execution_thread;
end a_process.two_threads;

Exemple d’extension de composant 

•  a_process.two_threads conAent deux threads 
Les features— les ports 

•  Les ports modélisent les échanges d’informaAon.  
  data : transport de données ; comme dans un circuit électronique 
  event : émission d’un signal 
  event data : signal + données ; comparable à un message 

•  les ports peuvent être déclarés en 
  entrée (in) 
  sorAe (out) 
  entrée‐sorAe (in out) 

•  Contrairement aux data ports, les event ports peuvent être mis 
en file d’acente 
  L’informaAon envoyée sur un port data peut écraser la précédente 
  La longueur de la file peut être spécifié par l’intermédiaire d’une 

propriété AADL 
  Si la file d’acente est pleine, une des poliAque suivantes peut être 

adoptée: 
o DropOldest: On supprime le message le plus ancien 
o DropNewest: On supprime le message le plus récent 
Features: paramètres & sous‐programmes d’interface 

•  Pour des sous‐programmes, on parle de paramètres 
(parameters) 

•  Un paramètre s’uAlise comme un port de donnée  
 data port 
 event data port 

•  Les paramètres peuvent être in, out ou in out 

•  Un composant thread ou data peut offrir des sous‐programmes 
comme interface 
 un thread serveur  

o p.ex. dans le cas d’un appel de procédure distante (RPC) 
 un composant de donnée proposant des méthodes d’accès 

o analogie avec les classes des langages objets 
Features: les accès aux composants 

•  Un composant peut indiquer qu’il requiert (requires) ou qu’il 
fournit (provides) un accès à un sous‐composant 
 un bus, p.ex. pour un processor ou une memory 

 une data, p.ex. pour une donnée partagée entre plusieurs threads 

•  On exprime ainsi l’obligaAon de brancher un composant avec un 
autre pour obtenir un système cohérent 
 un processor, une memory ou un device doivent être connectés aux 

autres composants matériels par l’intermédiaire d’un bus 
Features: les groupes de ports 

•  Regroupement de ports associés entre eux 
 facilite la manipulaAon au niveau de la descripAon 
 analogie avec un câble 
data pressure
end pressure;

data altitude
end altitude;

thread altimeter
features
  P : in event data port pressure;
  A : out data port altitude;
end altimeter;

thread client_altitude
features
  A : in data port altitude;
end client_altitude;

device pressure_sensor
features
  P : out event data port pressure;
end pressure_sensor;

subprogram compute_altitude
features
  P : in parameter pressure;
  A : out parameter altitude;
end compute_altitude;

Exemple de features 
Les connexions 

•  Pour relier les « features » entre elles 
 ports 
 paramètres 
 sous‐programmes d’interface 
 accès aux sous‐composants 
 groupes de ports 

•  Les connexions ont une direcAon 
  les entrées sont reliées aux sorAes de sous‐composants 
  les entrées d’interfaces sont reliées aux entrées des sous‐composants ; 

idem pour les sorAes 
•  Les features de sorAe peuvent être « 1 vers n » 
•  event data ports & event ports entrants 

 « n vers 1 » car gesAon de files d’acente 
•  les autres features entrantes 

 « 1 vers 1 » 
process manager
features
  P : in event data port pressure;
end manager;

thread implementation altimeter.basic
calls {
  appli : subprogram compute_altitude;
};
connections 
  parameter P -> appli.P;
  parameter appli.A -> A;
end altimeter.basic;

process implementation manager.altitude 
subcomponents
  client1 : thread client_altitude;
  server1 : thread altimeter.basic;
connections 
  pressure_input : event data port P -> server1.P;
  data port server1.A -> client1.A;
end manager.altitude; 

Exemple de connexion (1) 
data signal
end signal;

port group signal_DB9
features 
  CD  : in  data port signal;   -- carrier detection
  RD  : in  data port signal;   -- data reception
  TD  : out data port signal;   -- data transmission
  DTR : out data port signal;   -- ready to transmit data
  DSR : in  data port signal;   -- ready to send data
  RTS : out data port signal;   -- transmision request
  CTS : in  data port signal;   -- ready for transmission
  RI  : in  data port signal;   -- reception indicator
end signal_DB9;

port group signal_DB9_inverse inverse of signal_DB9
end signal_DB9_inverse;

Exemple de connexion (2a) 
system serial_card 
features
  plug : port group signal_DB9;
end serial_card;

system serial_wire
features
  plug : port group signal_DB9_inverse;
end serial_wire;

system global
end global;

system implementation global.basic
subcomponents
  card : system serial_card;
  wire : system serial_wire;
connections
  port group card.plug -> wire.plug;
end global.basic;

Exemple de connexion (2b) 
data a_data
end a_data;

subprogram prog1
features
  input : in parameter a_data;
  output : out parameter a_data;
end prog1;

subprogram prog2
features
  input : in parameter a_data;
  output : out parameter a_data;
end prog2;

subprogram implementation prog2.impl
calls { 
  a_call : subprogram prog1;
}; 
connections 
  parameter input -> a_call.input;
  parameter a_call.output -> output;
end prog2.impl; 

Exemple de connexion (3) 
Les modes 

•  Les modes permecent de modéliser la reconfiguraAon du 
système en foncAon d’événements 
 définis au niveau des composants 

 seuls les événements (event ports) peuvent déclencher un changement 
de mode 

 certains sous‐composants, connexions, etc. ne sont acAvés que dans 
certains modes 

•  Les changements de modes permecent de représenter des 
architectures dynamiques 
 évoluAon dans la configuraAon de l’architecture 

 ensemble déterminé de configuraAons possibles 
thread execution_thread
end execution_thread;

process a_process
features
  multi_thread : in event port;
  mono_thread : in event port;
end a_process;

process implementation a_process.configurable
subcomponents
  thread_1 : thread execution_thread;
  thread_2 : thread execution_thread in modes (multitask);
modes
  monotask : initial mode;
  multitask : mode;
  monotask -[ multi_thread ]-> multitask;
  multitask -[ mono_thread ]-> monotask;
end a_process.configurable; 

Exemple de modes 
Les flots (1) 

•  Les flots permecent de matérialiser les chemins à travers 
les éléments d’une descripAon. Ils suivent plus ou moins les 
connexions. 
 ne modélisent rien de concret 
 Une forme de redondance pour pouvoir analyser plus facilement 
les flots de données et y associer des propriétés 

•  on peut décrire : 
 le début d’un flot (flow source) 
 le milieu d’un flot (flow path) 
 la fin d’un flot (flow sink) 
 un flot complet (end to end flow) 
Les flots (2) 

•  Les component types conAennent  
 des flow specificaAons (source, sink et path) qui ne font intervenir 
que des features 

•  les component implementa%ons conAennent 
 des flow implementaAons (source, sink et path) qui font intervenir 
des features, des connecAons et des flots de sous‐composants 

 des end to end flows qui commencent par un flow source et 
finissent par un flow sink. 
process foo
features
  InitCmd : in  event      port;
  Signal  : in        data port signal_data;
  Result1 : out       data port position.radial;
  Result2 : out       data port position.cartesian;
  Status  : out event      port;
flows
  Flow1 : flow path   Signal -> Result1;
  Flow2 : flow path   Signal -> Result2;
  Flow3 : flow sink   InitCmd;
  Flow4 : flow source Status;
end foo;

Exemple de flow specificaAon 
Exemple de flow implementaAon 

process implementation foo.basic
subcomponents
A: thread bar.basic;
-- bar has a flow path fs1 from port p1 to p2
-- bar has a flow source fs2 to p3
C: thread baz.basic;
B: thread baz.basic;
-- baz has a flow path fs1 from port p1 to p2
-- baz has a flow sink fsink in port reset
connections
Conn1        : data  port signal  -> A.p1;
Conn2        : data  port A.p2    -> B.p1;
Conn3        : data  port B.p2    -> result1;
Conn4        : data  port A.p2    -> C.p1;
Conn5        : data  port C.p2    -> result2;
Conn6        : data  port A.p3    -> status;
connToThread : event port initcmd -> C.reset;
flows
Flow1: flow path signal -> conn1 -> A.fs1 -> conn2 -> B.fs1 -> conn3 -> result2;
Flow2: flow path signal -> conn1 -> A.fs1 -> conn4 -> C.fs1 -> conn5 -> result2;
Flow3: flow sink initcmd -> connToThread -> C.fsink;
-- a flow source may start in a subcomponent,
-- i.e., the first named element is a flow source
Flow4: flow source A.fs2 -> connect6 -> status;
end foo.basic;

Les propriétés en AADL (1) 

•  Les propriétés peuvent être associées à quasiment tous les 
éléments d’une descripAon 

• Une propriété permet d’associer une valeur d’un certain 
type (ou non typée) à un nom. 
 la norme prévoit un ensemble de propriétés standard 

 il est possible de définir de nouvelles propriétés dans des 
ensembles de propriétés (property sets) 

• Une propriété peut ne s’appliquer qu’à un ensemble de 
catégories d’éléments (p.ex. les processeurs) 
Les propriétés en AADL (2) 

•  Le type d’une propriété peut être 
 un booléen : aadlboolean 
 un enAer : aadlinteger 
 un réel : aadlreal 
 une chaîne de caractères : aadlstring 
 une énuméraAon : enumeration 
 une catégorie d’élément : classifier (composant, connexion, 
etc.) 

 une référence à un element: reference (composant…) 
 une plage de valeurs : list of … 
Supported_Queue_Processing_Protocols : 
  type enumeration (FIFO, <project_related>);

Queue_Processing_Protocol : 
  Supported_Queue_Processing_Protocols => FIFO
    applies to (event port, event data port, subprogram);

Source_Text : inherit list of aadlstring
  applies to (data, port, subprogram, thread, thread group,
              process, system, memory, bus, device, processor,
              parameter, port group);

Max_Thread_Limit : constant aadlinteger => <project_related>;

Thread_Limit : aadlinteger 0 .. value (Max_Thread_Limit) 
  => value (Max_Thread_Limit) applies to (processor);

Exemples de propriétés prédéfinies 
Les propriétés en AADL (3) 

•  Pouvant dépendre d’un mode 
•  Pouvant se propager dans les sous‐composants 

•  mot‐clef inherit 
•  Pouvant s’ajouter ou remplacer la valeur définie dans un composant 

père 
•  extension de composant 

•  Pouvant être déclarées  
•  dans un composant,  
•  au niveau de la déclaraAon d’un sous‐composant, une connexAon 
•  dans un composant père et s’appliquer à un de ses sous‐élément (sous‐

composant, feature, appel à un sous‐programme…) 

•  les propriétés permecent (entre autres) d’indiquer le déploiement 
des composants logiciels sur les composants matériels 
processor a_processor 
end a_processor;

processor implementation a_processor.simple
properties
  Thread_Swap_Execution_Time => 0ms .. 10 ms;
end a_processor.simple;

process a_process
end a_process;

system global
end global; 

system implementation global.simple
subcomponents
  processor1 : processor a_processor.simple {Thread_Limit => 3;};
  process1   : process a_process;
properties
  Actual_Processor_Binding => reference processor1 applies to process1;
end a_process.simple;

Exemples d’associaAons de propriétés 
property set our_properties is
  pressure     : type aadlinteger units (Pa, HPa => 100 * Pa);
  --  Définition des unités Pascal et Hecto-Pascal

  pressure_max : pressure applies to (device);
end our_properties;

system a_system
end a_system;

device a_sensor
end a_sensor;

system implementation a_system.with_a_sensor
subcomponents
  the_sensor : device a_sensor 
    {our_properties::pressure_max => 1020 hPa;}; 
end a_system.with_a_sensor;

Exemple d’ensemble de propriétés 
thread Collect_Samples
features
  Input_Sample   : in  data port Sample;
  Output_Average : out data port Sample;
annex OCL {** 
  pre:  0 < Input_Sample < maxValue;
  post: 0 < Output_Sample < maxValue;
**};
end Collect_Samples;

Les annexes 

•  Permecent d’enrichir la descripAon en uAlisant une syntaxe 
autre que celle d’AADL 
 p.ex. Z, OCL, etc. 

•  Peuvent être contenues dans les composants et les groupes 
de ports 
Annexes et propriétés 

•  Deux façons d’enrichir une descripAon AADL 
  uAliser des annexes 
  uAliser des propriétés 

•  Une annexe n’est pas obligatoirement interprétée par les ouAls 
d’exploitaAon 
  mais sa détecAon par un ouAls qui ne peut pas l'interpréter NE DOIT PAS causer des 

erreurs de syntaxe. 
•  On peut associer des propriétés à quasiment tous les éléments d’une 

descripAon 
•  Une annexe sert à ajouter des précisions facultaAves. Une propriété est 

généralement très liée à la descripAon architecturale 
•  Les annexes et propriétés permecent une grande souplesse 

  acenAon à ne pas détourner leur uAlisaAon 
o  décrire tout le comportement d’un sous‐programme avec des annexes au lieu d’uAliser les 

séquences d’appel 
o  uAliser une propriété pour référencer un fichier qui conAent la descripAon en langage naturel 

de l’architecture d’un composant 
OrganisaAon d’une descripAon 

•  Les paquetages (packages) 
  matérialisent des espaces de nom 

  une parAe publique : visible de partout 
  une parAe privée : uniquement visible depuis le paquetage 

o  composants 

o  groupes de ports 
o  Propriétés 

•  Espace de nom anonyme (anonymous namespace) 
  le plus haut niveau de la descripAon 

o  paquetages 
o  composants 

o  groupes de ports 
o  ensembles de propriétés 

•  Les systèmes permecent de structurer l’architecture 

•  Les paquetages permecent de structurer la descripAon 
package machines 
public 
  system a_machine
  end a_machine;

  system implementation a_machine.mono_processor 
  subcomponents 
    the_processor : processor machines::elements::a_processor.specific;
  end a_machine.mono_processor;
private
  system a_private_machine
  end a_private_machine;
end machines;

package machines::elements
public
  processor a_processor
  end a_processor;

  processor implementation a_processor.specific
  end a_processor.specific;
end machines::elements;

Exemple d’uAlisaAon des paquetages 
system global_system
end global_system;

system implementation global_system.two_machines
Subcomponents
  machine_1 : system machines::a_machine.mono_processor;
  machine_2 : system machines::a_machine;
end global_system.two_machines;

DescripAon complète 

•  Une modélisaAon AADL est un ensemble de déclaraAons de composants 

•  Les sous‐composants sont des instances des déclaraAons 

•  Un système doit jouer le rôle de racine pour l’architecture 
  pas d’interface 
  une implémentaAon contenant les sous‐composants 
Synthèse des sous‐éléments (1) 

•  component types 
  features 
  flows 
  properAes 
  annexes 

•  port groups 
  features (et/ou inverse of) 
  properAes 
  annexes 

•  component implementaAons 
  subcomponents 

  calls 
  connecAons 
  flows 
  modes 

  properAes 
  annexes 
Synthèse des sous‐éléments (2) 

•  properAes : tous les composants 

• modes : tous les composants 
•  flows : les composants d’exécuAon 

 subprogram, thread, thread group, process, processor, device, 
system 

•  connecAons : tous les composants qui ont des sous‐
composants 

•  subprogram calls : thread, subprogram 
•  features : propres à chaque type de composant 
InstanciaAon de l’architecture 

• Une descripAon AADL est une suite de déclaraAons 
 analogie avec un diagramme de classes UML 
 analyse et manipulaAon limitées 

o certaines propriétés sont associées à un sous‐composant parAculier 

•  Il est nécessaire d’instancier la modélisaAon 
 arborescence d’instances de composants correspondants aux 
déclaraAons de sous‐composants 
o Valeurs de propriétés résolues  

 les enAtés qui ne sont pas instanciées: 
o les composants data associés aux features car il représentent un type. 

 on manipule alors l’architecture telle qu’elle doit être dans la 
réalité 
GénéraAon d’un système exécutable 

•  Une modélisaAon AADL décrit 
  les caractérisAques des composants 
  les connexions 

•  Les propriétés standard permecent d’associer un code source à 
chaque composant 
 VHDL, … 
 Ada, C, … 

•  Ces codes sources indiqués en propriétés doivent se conformer 
aux spécificaAons AADL (signatures des sous‐programmes…) 
 écrits à la main 
 générés automaAquement par d’autres ouAls 

•  Une annexe du standard donne les correspondances entre les 
syntaxes AADL et Ada ou C 

•  On peut indiquer les codes sources des composants et de 
générer une applicaAon pour un exécuAf AADL 
Paramètres pour la généraAon 

•  Composants 
 matériels : fournissent les informaAons de déploiement 

o caractérisAques des machines 
o connexions sur les réseaux 
o … 

  logiciels 
o correspondent aux applicaAons dont il faut générer le code 
o processus : modélisent les nœuds (parAAons) 
o threads : éléments acAfs 
o sous‐programmes : éléments réacAfs des applicaAons 

 systèmes 
o éléments de structure non foncAonnels de l’architecture 

•  Interfaces et connexions entre les processus 
 correspondent aux modèles de réparAAon uAlisés 
UAlisaAon d’un intergiciel 

•  ExécuAf AADL = intergiciel 
 thread AADL fourni par l’intergiciel ou le noyau 
 communicaAons assurées par l’intergiciel 

•  Générateur de code: 
 généraAon de code applicaAf à parAr des données et des sous‐

programmes AADL 
 généraAon de code intergiciel pour l’adapter à l’applicaAon AADL 

o  interfaçage des threads et des processus AADL 
o traducAon des mécanismes de communicaAon en foncAon de l’intergiciel 

•  Les composants applicaAfs sont pilotés par l’exécuAf 
 code uAlisateur encapsulé, n’a aucun contrôle sur l’exécuAf 
 possibilité de vérificaAon 

o à condiAon de connaître les correspondances entre un thread AADL et la 
configuraAon correspondante de l’intergiciel 

  le contrôle par l’intergiciel limite la liberté de modélisaAon 
o certaines construcAons architecturales ne sont pas possibles 
ExécuAf pour les nœuds applicaAfs 

•  L’exécuAf AADL doit assurer deux tâches 
 contrôle de l’exécuAon des threads (ordonnancement) 
 prise en charge des communicaAons (inter‐ et intra‐ nœuds) 

intergiciel 

couche d’encapsulation 

appli. appli. appli. 

Composants générés 
automatiquement à partir 
de la description AADL

Descriptions fonctionnelles 
(fournies par lʼutilisateur)

Intergiciel Minimal 

VérificaAon de la descripAon AADL 

•  VérificaAon syntaxique 
 valeur des propriétés bornées 
 ensemble de propriétés AADL_Project caractérisAque des dimensions de 

l’exécuAf 
o protocoles possibles, etc. 

•  VérificaAon de la cohérence des spécificaAons 
 tailles mémoire des données et des mémoires 
 temps d’exécuAon des sous‐programmes et des threads 
 nombre de threads par processeur 

•  VérificaAon des limitaAons relaAves à l’exécuAf 
 bonne uAlisaAon des propriétés vis‐à‐vis des capacités de l’exécuAf 
 bonne uAlisaAon des modes 
VérificaAon des flux d’exécuAon 

•  VérificaAon staAque des connexions 
• UAlisaAon de méthodes formelles 

•  p.ex.: réseaux de Petri 
 pour étudier l’absence d'interblocages provoqués par l’assemblage 
des composants 

 pour vérifier la définiAon correcte des données 

 détecter les sous‐ensembles architecturaux jamais uAlisés 
 … 
VérificaAon de l’ordonnançabilité 

•  Au niveau des processus et des processeurs 
 contrainte sur le nombre maximum de threads 
 calcul de l’ordonnancement 

•  ExploitaAon des propriétés AADL 
 protocole d’ordonnancement 

 temps d’exécuAon 
 communicaAons entre les nœuds 

•  p.ex.: Cheddar 
GénéraAon de l’applicaAon 

•  TraducAon des composants AADL en code source 
 types de données 
 Procédures 

•  ConfiguraAon de l’exécuAf en foncAon de l’architecture 
 nombre de threads 
 entrées/sorAes 
 périodes 
 … 

•  AdéquaAon entre le code généré et la modélisaAon 
architecturale 
 les étapes de vérificaAon et la généraAon de code doivent reposer 
sur les mêmes spécificaAons d’exécuAf 
Ocarina: un ensemble d’ouAls AADL 

•  Bibliothèque & ouAls pour 
manipuler AADL  
  Parseurs & afficheurs AADL 
  VérificaAon sémanAque 

•  Générateurs de code 
  Ada/PolyORB 
  (Ada, C)/PolyORB‐HI 
  Pour plusieurs plateformes (NaAve, 

LEON, ERC32) 
•  ConfiguraAon du support 

d'exécuAon 
•  VérificaAon & ValidaAon 

  Réseaux de Petri 
 Ordonnancement (Cheddar) 

•  Un ouAl en ligne de commande 
pour automaAser ces tâches 

Core  
Library 

AADL Model 

Specific Analyzers 
(Hardware checks…) 

Model 
Transformation 

Code 
Generators 

Petri Nets 
(CPN-AMI) 

Sched. Analysis 
(Cheddar) 

Ada / C AADL 

Runtime 

Specific Analyzers 
(Hardware checks…) 

Runtime 

PolyORB‐HI: Intergiciel pour les systèmes criAques 

•  Contraintes temps réel dur spécifiques aux systèmes criAques: 
 Modèle de concurrence analysable : Profil Ravenscar 
  RestricAons du langage de programmaAon pour les systèmes criAques 

o  Encore plus restricAf que le profil Ravenscar 
  Pas d’allocaAon dynamique ni d'orienté objet 

•  PolyORB‐HI: un support d'exécuAon AADL 
  Supporte les construcAons AADL 

o  Threads périodiques et sporadiques, données, etc. 
  Configure automaAquement à parAr du modèle AADL 

o  Ressources calculées et allouées staAquement 
o  Pas d’intervenAon requise de la part de l’uAlisateur 

 Occupe une faible taille en mémoire 
o  Toute la valeur ajoutée est dans la phase de généraAon de code 

  Contribuer à la thémaAque des “usines à intergiciels” 
Conclusions 

•  ModélisaAon concrète 
  dernière phase avant la généraAon/déploiement du système 
  précision dans la modélisaAon 
  exploitaAon pour la vérificaAon et la généraAon de prototypes 
  possibilités de correcAons sur l’architecture 

•  AADL offre une grande souplesse de modélisaAon 
  degré de modélisaAon selon les besoins 
  peut être uAlisé comme langage fédérateur 

o  exploitaAon par plusieurs ouAls différents 

•  hcp://aadl.telecom‐paristech.fr/ocarina 
•  hcp://aadl.telecom‐paristech.fr/polyorb‐hi 

•  hcp://beru.univ‐brest.fr/~singhoff/cheddar/index‐fr.html 
•  hcp://www‐src.lip6.fr/logiciels/mars/CPNAMI/  


